Supplementary Appendix – WinBUGS code for Full model
model {
# Duration 
# Duration of asymptomatic infection

dura ~ dnorm(0,0.0001) I(0,)  
D ~ dnorm(dura,pr.D) 

dev.D <- pr.D*pow((dura-D),2)  # Duration Deviance
lambdaC <- 1/dura              # clearance rate

# Duration of symptomatic infection
durs ~ dunif(0.0767,0.1533)  # symptomatic, 4-8 weeks

#durs ~ dunif(0.0577,0.2308) #symptomatic, 3-12 weeks sensitivity analysis

cut.durs <- cut(durs)        # pevents updating of durs

lambdaS <- 1 / cut.durs      # clearance rate  
# Probability ct is symptomatic
r.phi ~ dbin(phi,n.phi)  phi ~ dbeta(1,1) 
dev.phi <- 2 * (r.phi * log(r.phi / (phi *  n.phi)) +  (n.phi - r.phi) *                                 

           log((n.phi - r.phi) / (n.phi - (n.phi * phi))))
# Deviance
# Mean duration of infection

dur <- phi*cut.durs + (1-phi)*dura

# Incidence model

# Likelihood and model
for (i in 1:2) { # loop over rate, 1=infection, 2=reinfection

 for (s in 1:3) { # loop over setting - GP=1, fP=2, GUM=3

  for (a in 1:3) { # loop over age group:1=16,17;2=18-20, 3=21-24

   lambda[i,s,a] <- equals(i,1)*(gamma[a] * rho[s] * lambda111) +




  equals(i,2)*(gamma[a] * rho[s] * lambda111 * eta[s])

   r.inc[i,s,a] ~ dbin(theta.inc[i,s,a],n.inc[i,s,a])

   theta.inc[i,s,a] <- phi * (

                       (1 - (lambdaS + lambda[i,s,a] *

                        exp(-(lambdaS + lambda[i,s,a]) * 0.5)) /                                                         

                       (lambdaS + lambda[i,s,a]))




     ) + (1 - phi) * (  




     (1 - (lambdaC + lambda[i,s,a] *




     exp(-(lambdaC + lambda[i,s,a]) * 0.5)) /                                        
                       (lambdaC + lambda[i,s,a]))




                     )

  dev.inc.rat[i,s,a] <- 2 * (r.inc[i,s,a] * log(r.inc[i,s,a] /                                              

                        (theta.inc[i,s,a] * n.inc[i,s,a])) + (n.inc[i,s,a]-                             

                        r.inc[i,s,a]) * log((n.inc[i,s,a] - r.inc[i,s,a]) /                           
                       (n.inc[i,s,a] - (n.inc[i,s,a] * theta.inc[i,s,a])))) 

   }

  }

 }

sumdev.inc.rat <- sum(dev.inc.rat[ , , ])   # Deviance, # LaMontagne

#priors for log incidence parameters
loglambda111 ~ dnorm(0,.0001)                  #Age 16-17, GP, infection

log(lambda111) <- loglambda111

for (s in 1:3) {
 logeta[s] ~ dnorm(0,.0001)     # GP=1, fP=2, GUM=3

 log(eta[s]) <- logeta[s]
 }

logrho[1] <- 0                  # GPtheta.inc

for (s in 2:4) {
 logrho[s] ~ dnorm(0,.0001) # 2=fp, 3=GUM, 4=pop, rel to GP

 }    
for (s in 1:4) {
 log(rho[s]) <- logrho[s]
 }

loggamma[1] <- 0            #1 = 16-17

for (a in 2:5) {
 loggamma[a] ~ dnorm(0,.01) #2=18-19, 3=20-24,4=25-29,5=30-44 rel to 16-17

}    

for (a in 1:5) {
 log(gamma[a]) <- loggamma[a]
 }   

# setting specific Odds ratios from Adams
for (s in 1:3) {                            #s+1= 2=FP, 3=GUM, 4=pop, rel to GP

   r.OR[s] ~ dnorm(logrho[s+1],pr.OR[s]) 

   dev.OR[s] <- pr.OR[s] * pow((r.OR[s]-logrho[s+1]),2)  # OR deviance

}

sumdev.OR <- sum(dev.OR[])

# overall age-specific incidence in population, 1=16-17, 2=18-19, 3=20-24,

#                                               4=25-29, 5=30-44
for (a in 1:3) {
 p[a] ~ dbeta(aprior[a],bprior[a])

 cut.p[a] <- cut(p[a])                      # prevents updating p[a]
 lamda.F[a] <- rho[4]*((1-cut.p[a])*lambda[1,1,a] + 

               cut.p[a]*lambda[2,1,a])

 lamda.pop[a] <- lamda.F[a] / (1 - lamda.F[a] * dur)

 }

for (a in 4:5) { 
 lamda.pop[a] <- lamda.pop[1] * gamma[a]
 }

# Age-specific Prevalence from Adams, 1=18-19, 2=20-24, 3=25-29, 4=30-44
for (a in 1:4) {
 Lprev[a]~dnorm(lprev[a+1],pr.Lprev[a])

 dev.prev[a] <- pr.Lprev[a] * pow((Lprev[a]-lprev[a+1]),2)

 }

for (a in 1:5) {
 lprev[a] <- logit(prev[a])
 }

for (a in 1:3) {
 prev[a] <- min(.999,(dur * lamda.pop2[a])) 
}
prev[4] <- prev[3] * gamma[4]/gamma[3]

prev[5] <- prev[3] * gamma[5]/gamma[3]

sumdev.prev <- sum(dev.prev[])

sumdev <- dev.D + sumdev.inc.rat + sumdev.OR + sumdev.prev + dev.phi

# incidence in difference age bands, using population weights N[]
for (a in 16:17) {inc[a] <- lamda.pop[1]}

for (a in 18:20) {inc[a] <- lamda.pop[2]}

for (a in 21:24) {inc[a] <- lamda.pop[3]}

for (a in 25:29) {inc[a] <- lamda.pop[4]}

for (a in 30:44) {inc[a] <- lamda.pop[5]}

lamda.pop2[1] <- lamda.pop[1]  # age 16,17

lamda.pop2[2] <- lamda.pop[2]  # age 18,19. 

lamda.pop2[3] <- inprod(inc[20:24],N[20:24])/sum(N[20:24])  # age 20-24

inc1624 <- inprod(inc[16:24],N[16:24])

inc2544 <- inprod(inc[25:44],N[25:44])

inc1644 <- inprod(inc[16:44],N[16:44])

lamda.pop2[4] <- inc1624 / sum(N[16:24]) # age 16-24

lamda.pop2[5] <- inc2544 / sum(N[25:44]) # age 25-44

lamda.pop2[6] <- inc1644 / sum(N[16:44]) # age 16-44 

# other population summaries
prop.treat.1624.2002 <- 31510 / inc1624

prop.treat.1624.2003 <- 34660 / inc1624

prop.asymp.clin.1624.2003 <- 1 - (phi / prop.treat.1624.2003)

# prevalence in reconstructed age groups
for (a in 1:6) {
 prev.pop2[a] <- dur * lamda.pop2[a] 
 }

} # end of program
Data

list(
# duration
D=1.36, pr.D=59.17,    # from Duration paper 2-Class estimate 

# incidence 
r.inc=structure(.Data=c(

4,3,4, 9,5,7, 5,16,9, 

5,7,10, 13,12,5, 6,15,5

),.Dim=c(2,3,3)), 
n.inc=structure(.Data=c(

73,195,188, 194,273,201, 102,235,245, 
14,65,79, 95,127,63, 40,139,81

),.Dim=c(2,3,3)),

# priors for infection/reinfection weights, 16-17, 18-20, 21-24

aprior=c(46.5,73.1,106.4), bprior=c(400,634.6,938.4),  
# Logit Prevalence 
Lprev=c(-2.987,-3.41,-4.185,-4.82),

pr.Lprev=c(19.3,20.8,18.4,17.2),

# Setting-specific Log Odds Ratios. Fp,GUM, pop

r.OR= c(0.239,0.871, -.511), 

pr.OR= c(67.03,35.21, 17.28),

# symptomatic ct - Geisler

r.phi = 26, n.phi = 115,

# Population sizes from census, age =1...44 - 2002
N=c(NA,NA,NA,NA,NA, NA,NA,NA,NA,NA, NA,NA,NA,NA,NA, 

305500,306300,296400,291400,294800, 310100,313900,305600,294700,295000,

304100,317000,329600,349600,370300, 380900,376900,387800,390900,399400,

401200,402600,398700,391900,381900, 370900,356200,349000,343800)
)

# Initial values - 1
# Duration
list(dura=1,  durs=.115,    

# incidence

loglambda111=-2.5,

logeta=c(1,.5,.5),

logrho=c(NA, 1, 1.5,-.5), 

loggamma=c(NA,-.2,-.5,-.7,-.8), 
# probability symptomatic
phi = 0.5,

# proportion re-infection
p=c(.1, .1, .1)

)

# Initial values - 2
# Duration
list(dura=.2,  durs=.112,    

# incidence

loglambda111=-4,

logeta=c(.5,1,1),

logrho=c(NA, 0, 0,0), 

loggamma=c(NA,0,0,0,0), 
# probability symptomatic
phi = .1,

# proportion re-infection
p=c(.3, .05, .2)

)

